

```

10 ' -GUNBOATS-
20 ' D.MUIR '84
30 GOTO 1310
40 '
50 ' set up
60 '
70 MODE 1
80 BORDER 13
90 DEFINT a-r:DEFSTR t-z
100 DEF FNsa=ax*TAN(PI/6)-(((ax*ax)/(pv*pv*COS(PI/6)*COS(PI/6)))^2*16)
110 DEF FNsc=bx*TAN(PI/6)-(((bx*bx)/(cv*cv*COS(PI/6)*COS(PI/6)))^2*16)
120 DEF FNe(bb)=(bb-1)*16+6:DEF FNf(bb)=(25-bb)*16+7
130 SYMBOL 255,3,7,14,28,56,44,36,24
140 SYMBOL 254,192,224,112,56,28,52,36,24

```

```

150 SYMBOL 253,255,255,255,127,127,63,15
160 SYMBOL 252,255,255,255,255,254,254,252,240
170 DIM m$(1,1)
180 m$(0,0)=" "+CHR$(141)+CHR$(255)+" "
190 m$(0,1)=" "+CHR$(253)+CHR$(143)+CHR$(252)+" "
200 m$(1,0)=" "+CHR$(254)+CHR$(142)+" "
210 m$(1,1)=m$(0,1)
220 INK 0,23:INK 1,12:INK 2,9:INK 3,6
230 CLS
240 WINDOW #1,1,40,20,25:PAPER #1,2:PEN #1,0:CLS #1
250 WINDOW #2,10,20,21,21:PAPER #2,0:PEN #2,2:CLS #2
260 WINDOW #3,10,20,22,22:PAPER #3,0:PEN #3,2:CLS #3
270 WINDOW #4,10,20,23,23:PAPER #4,0:PEN #4,2:CLS #4
280 WINDOW #5,25,35,21,21:PAPER #5,0:PEN #5,2:CLS #5
290 WINDOW #6,25,35,23,23:PAPER #6,0:PEN #6,2:CLS #6
300 WINDOW #7,10,20,24,24:PAPER #7,0:PEN #7,2:CLS #7
310 PEN 1
320 FOR i=0 TO 1:PEN 1:LOCATE 5,18+i:PRINT m$(0,i);:PEN 3:LOCATE 29,18+i:PRINT m
$(1,i);:NEXT
330 nx=5:bc=0:bm=0:cx=29:num=0:pfk=0:cv=25:score=0:pli=5:pv=25:cli=3:cr=0:bo=
0
340 LOCATE #1,1,2:PRINT #1,"VELOCITY"
350 PRINT #1,"SCORE"
360 PRINT #1,"LIVES"
370 PRINT #1,"BOATS"
380 GOSUB 770:GOSUB 780:GOSUB 790:GOSUB 800:GOSUB 810:GOSUB 820
390 '
400 ' player decision
410 '
420 EVERY 350,3 GOSUB 960
430 EVERY 200,2 GOSUB 690
440 EVERY 40,1 GOSUB 610
450 IF lck THEN lck=0:FOR i=1 TO 500:NEXT:GOTO 420
460 IF k1 THEN 1260
470 EI:DI
480 IF INKEY(74)=0 OR INKEY(8)=0 THEN nx=nx-1*(nx=1):GOSUB 570:GOTO 450
490 IF INKEY(75)=0 OR INKEY(1)=0 THEN nx=nx+1*(nx=15):GOSUB 570:GOTO 450
500 IF INKEY(72)=0 OR INKEY(0)=0 THEN pv=pv+1*(pv>30):GOSUB 770:GOTO 450
510 IF INKEY(73)=0 OR INKEY(2)=0 THEN pv=pv-1*(pv<20):GOSUB 770:GOTO 450
520 IF INKEY(76)=0 OR INKEY(9)=0 THEN GOSUB 860:GOTO 450
530 FOR i=1 TO 50:NEXT:GOTO 450
540 '
550 ' print player new position
560 '
570 PEN 1:FOR i=0 TO 1:LOCATE nx,18+i:PRINT m$(0,i);:NEXT:RETURN
580 '
590 ' computer boat move
600 '
610 DI:IF NOT bm THEN EI:RETURN
620 num=num+1:IF num>bc THEN bm=0:num=0:EI:RETURN
630 IF cx+dir>35 OR cx+dir<21 THEN bm=0:num=0:EI:RETURN
640 cx=cx+dir
650 PEN 3:FOR i=0 TO 1:LOCATE cx,18+i:PRINT m$(1,i);:NEXT:EI:RETURN
660 '
670 ' set computer boat move
680 '
690 DI:IF bm THEN EI:RETURN
700 bm=-1
710 dir=INT(RND*2)*2-1
720 bc=4+INT(RND*8)
730 EI:RETURN
740 '
750 'update scoreboards subroutines
760 '
770 CLS #2:PRINT #2,pv;:RETURN
780 CLS #3:PRINT #3,sc;:RETURN
790 CLS #4:PRINT #4,pli;:RETURN
800 CLS #6:PRINT #6,cli;:RETURN

```

```

CLS #5:PRINT #5,cv;:RETURN
820 CLS #7:PRINT #7,bo;:RETURN
830 '
840 ' player fire
850 '
860 psx=nx+5:psy=17:PEN 2:LOCATE psx,psy:PRINT CHR$(144);:opmx=psx:opmy=psy:ax=0
870 SOUND 7,500,10,4,0,0,5
880 ax=ax+1:ay=FNSa:SOUND 7,35-ay,15
890 IF psy-ay>19 OR psx+ax>40 THEN LOCATE opmx,opmy:PRINT " ";:RETURN
900 IF TEST(FNe(psx+ax),FNf(psy-ay))=3 THEN LOCATE opmx,opmy:PRINT " ";:GOSUB 11
40:RETURN
910 LOCATE opmx,opmy:PRINT " ";:opmx=psx+ax:opmy=psy-ay:LOCATE opmx,opmy:PRINT C
HR$(144);
920 GOTO 880
930 '
940 ' computer fire
950 '
960 DI:csx=cx:csy=17:PEN 2:LOCATE csx,csy:PRINT CHR$(144);:ocmx=csx:ocmy=csy:bx=
0
970 cv=cv+2*(cr-(cv<20)+(cv>30))
980 GOSUB 810
990 SOUND 7,500,10,4,0,0,5
1000 bx=bx+1:by=FNsc:SOUND 7,35-by,15
1010 IF csy-by>19 OR csx-bx<1 THEN cr=SGN(csx-bx-nx):LOCATE ocmx,ocmy:PRINT " ";
:E1:RETURN
1020 IF TEST(FNe(csx-bx),FNf(csy-by))=1 THEN LOCATE ocmx,ocmy:PRINT " ";:GOSUB 1
080:cr=0:E1:RETURN
1030 LOCATE ocmx,ocmy:PRINT " ";:ocmx=csx-bx:ocmy=csy-by:LOCATE ocmx,ocmy:PRINT
CHR$(144);
1040 GOTO 1000
1050 '
1060 'player boat hit
1070 '
1080 SOUND 7,900,50,7,0,0,7:r=12:FOR i=1 TO 10:r=15-r:INK 1,r:FOR j=1 TO 20:NEXT
:NEXT
1090 pli=pli-1:IF pli=0 THEN k1=-1
1100 GOSUB 790:RETURN
1110 '
1120 'computer boat hit
1130 '
1140 SOUND 7,900,50,7,0,0,7:r=6:FOR i=1 TO 10:r=6-r:INK 3,r:FOR j=1 TO 20:NEXT:N
EXT
1150 cli=cli-1:sc=sc+10-90*(cli=0):GOSUB 800:GOSUB 780
1160 IF cli>0 THEN RETURN
1170 pli=pli+1:GOSUB 790:cli=3:bo=bo+1:GOSUB 820
1180 FOR i=0 TO 1:LOCATE cx,18+i:PRINT " ";:NEXT:cx=29
1190 i=REMAIN(1):i=REMAIN(2):i=REMAIN(3):FOR i=1 TO 5:SOUND 7,700+i*50,50,i+2:NE
XT:FOR i=1 TO 1000:NEXT
1200 GOSUB 800
1210 PEN 3:FOR i=0 TO 1:LOCATE 29,18+i:PRINT m$(1,i);:NEXT
1220 lck=-1:RETURN
1230 '
1240 ' end of game
1250 '
1260 MODE 0:INK 1,12,24:INK 0,1:PAPER 0:PEN 1:PRINT " YOUR FINAL SCORE":PRINT:P
RINT " ";sc:PRINT:PRINT " ANOTHER GO ?"
1270 IF INKEY(46)=0 THEN END ELSE IF INKEY(43)=0 THEN RUN 70 ELSE 1270
1280 '
1290 ' titles
1300 '
1310 MODE 0:INK 1,12,24:INK 0,1:PEN 1:PAPER 0:CLS
1320 LOCATE 1,11
1330 PRINT " ";CHR$(143);CHR$(143);CHR$(133);CHR$(133);CHR$(138);CHR$(138);CHR$(
215);CHR$(133);CHR$(135);CHR$(215);CHR$(138);CHR$(143);CHR$(133);CHR$(214);CHR$(
215);CHR$(138);CHR$(143);CHR$(138)
1340 PRINT " ";CHR$(133);" ";CHR$(133);CHR$(138);CHR$(138);CHR$(213);CHR$(133);
CHR$(141);CHR$(212);CHR$(138);" ";CHR$(133);CHR$(133);CHR$(138);" ";CHR$(133);CH
R$(136)

```

```

1350 PRINT " ";CHR$(133);CHR$(213);CHR$(133);CHR$(133);CHR$(138);CHR$(138);" ";
CHR$(135);CHR$(215);CHR$(138);" ";CHR$(133);CHR$(143);CHR$(143);" ";CHR$(133)
1360 PRINT " ";CHR$(143);CHR$(143);CHR$(133);CHR$(213);CHR$(212);" ";CHR$(141)
;CHR$(212);CHR$(138);CHR$(143);CHR$(133);CHR$(133);CHR$(138)
1370 FOR i=1 TO 7:SOUND 7,50,i*10,i:FOR j=1 TO 50:NEXT:NEXT
1380 FOR i=1 TO 2000:NEXT
1390 CLS:PRINT "INSTRUCTIONS?"
1400 IF INKEY(46)=0 THEN RUN 70 ELSE IF INKEY(43)<>0 THEN 1400
1410 '
1420 ' instructions
1430 '
1440 MODE 1:INK 0,24:INK 1,1:CLS
1450 PRINT "INSTRUCTIONS":PRINT:PRINT "You control the gunboat on the left of":P
RINT "the screen; the computer that on the":PRINT "right.":PRINT:PRINT "Horizont
al joystick or cursor movement":PRINT "moves your ship."
1460 PRINT:PRINT "Vertical joystick or cursor movement":PRINT "raises and lowers
the muzzle velocity":PRINT "of your gun.":PRINT:PRINT "Use the fire button or c
opy key to shoot":PRINT
1470 PRINT "You have 5 lives and each enemy gunboat":PRINT "has three. You scor
e 10 points for":PRINT "each hit. 3 hits and you score 100 and":PRINT "gain an
extra life. Then a new enemy":PRINT "appears":PRINT
1480 PRINT:PRINT "(PRESS ANY KEY)";
1490 CALL &BB03
1500 u$=INKEY$:IF u$="" THEN 1500 ELSE RUN 70

```